Inverse Laplace Transforms and IVP's

```
> restart;with(inttrans);
```

$$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable]$$

Inverse Laplace Transform

The command for finding the inverse Laplace transform $f(t)$ of $F(s)$ is **invlaplace(F(s),s,t)**. We look at some examples. The following command assume puts an assumption on *b*, namely that $b > 0$. The variable *b* with the assumption is denoted by *b*~.

```
> assume(b,positive);
```

$F(s) = \dfrac{b}{s^2+b^2}$.

```
> invlaplace(b/(s^2+b^2),s,t);
```

$$\sin(b\sim t)$$

The following command takes the assumption off of *b*.

```
> b:='b';
```

$$b := b$$

$F(s) = \dfrac{s^2+1}{s^2\,(s+1)}$.

```
> invlaplace((s^2+1)/(s^2*(s+1)),s,t);
```

$$2\,e^{-t} - 1 + t$$

$F(s) = \dfrac{s^2+2}{(s^2+1)\,(s+1)}$.

```
> invlaplace((s^2+2)/((s^2+1)*(s+1)),s,t);
```

$$\frac{3}{2}\,e^{-t} - \frac{1}{2}\cos(t) + \frac{1}{2}\sin(t)$$

Differential Equations with the Laplace Transform

Let's consider the initial value problem $y'' - 2\,y' + 5\,y = -8\,e^{-t}$, $y(0) = 2$, $y'(0) = 12$.

```
> deq:=diff(y(t),t$2)-2*diff(y(t),t)+5*y(t)=-8*exp(-t);
```

$$deq := \frac{d^2}{dt^2}\,y(t) - 2\left(\frac{d}{dt}\,y(t)\right) + 5\,y(t) = -8\,e^{-t}$$

```
> IC:=y(0)=2,D(y)(0)=12;
```

$$IC := y(0) = 2,\ D(y)(0) = 12$$

We first solve the standard way. We use the infolevel command to give us more information about how Maple is carrying out the command in brackets. Setting **infolevel** to 5 gives the most information. The lowest and default level is 1.

```
> infolevel[dsolve]:=5;
```

$$infolevel_{dsolve} := 5$$

```
> dsolve({deq,IC},y(t));
Methods for second order ODEs:
--- Trying classification methods ---
trying a quadrature
trying high order exact linear fully integrable
trying differential order: 2; linear nonhomogeneous with symmetry [0,1]
trying a double symmetry of the form [xi=0, eta=F(x)]
-> Try solving first the homogeneous part of the ODE
   checking if the LODE has constant coefficients
   <- constant coefficients successful
   -> Determining now a particular solution to the non-homogeneous ODE
      trying a rational particular solution to g^(-1)*L*g = 1
      <- rational particular solution to g^(-1)*L*g = 1 successful
<- solving first the homogeneous part of the ODE successful
```

$$y(t) = 4\, e^{t} \sin(2\, t) + 3\, e^{t} \cos(2\, t) - e^{-t}$$

To have Maple solve a linear initial value problem using the Laplace transform, we include the option **method=laplace**.

```
> dsolve({deq,IC},y(t),method=laplace);
dsolve/inttrans/solveit: Transform of eqns is {_s1^2*laplace/internal(y
(t) t _s1)-8-2*_s1-2*_s1*laplace/internal(y(t) t _s1)+5*
laplace/internal(y(t) t _s1)+8/(_s1+1)}
dsolve/inttrans/solveit: Algebraic Solution is {laplace/internal(y(t) t
_s1) = 2*_s1*(_s1+5)/((_s1+1)*(_s1^2-2*_s1+5))}
```

$$y(t) = e^{t}\,(3\cos(2\, t) + 4\sin(2\, t)) - e^{-t}$$